

## ВСТУП

В наш час існує велика кількість різних математичних систем, які дозволяють користувачу проводити різні науково-технічні обчислення. Це і Mathematica, Maple, Derive, Origine, Mathlab, Mathcad і т.д. Кожна з них має свої переваги та недоліки. Якщо вибирати за двома критеріями за простотою освоєння і широтою можливостей, то однією з найкращих є система Mathcad. Саме їй і присвячена дана методична розробка. Вона буде корисною для студентів математичного, фізичного та інженерно-технічного факультетів.

Охопити всі можливості математичної системи Mathcad в даній методичній розробці неможливо. Тому в методичній розробці з курсу за вибором “Математичне забезпечення персонального комп’ютера” зосереджено увагу на основних можливостях, що пов’язані з предметами, якими студенти уже володіють. Це і інтегральне та диференціальне числення, знаходження границь, знаходження коренів многочлена, розв’язання систем лінійних та нелінійних рівнянь, знаходження максимальних та мінімальних значень функції, елементи математичної статистики та теорії ймовірності, розв’язання задач з інформатики і т.д.

Метою даного курсу є вироблення навиків у роботі з математичними системами та уміння розв’язувати науково-технічні задачі.

# 1. Загальні можливості математичної системи Mathcad

## 1.1. Загальні відомості

Математична система Mathcad містить дві досить поширені панелі – стандартну панель інструментів та панель форматування, а також містить нетипову для інших систем – панель виводу палітр математичних знаків (див. рис.).



Рис. Панель математичних знаків

Як видно з малюнку панель виводу палітр математичних знаків складається із дев'яти піктограм, кожна з яких визиває своє унікальне віконце.




Рис. Палітри математичних символів та операторів.

Розглянемо призначення кожної із палітр:

*Calculator* – палітра арифметичних операцій та елементарних функцій;

*Programming* – палітра засобів програмування;

*Greek* – палітра грецьких букв;

*Symbolic* – палітра ключових слів та операторів символічного обчислення;

*Boolean* – палітра логічних операцій та знаків відношення;

*Evaluation* – палітра знаків присвоєння та виводу результатів, шаблонів операторів типу Prefix, Postfix, ...

*Graph* – палітра шаблонів графіків різного типу;

*Matrix* – палітра шаблонів векторів і матриць та дій над ними;

*Calculus* – палітра операторів похідних, інтегралів, сум, добутків та інших.

Математична система Mathcad має широкі можливості для розв'язання науково-технічних задач, серед яких слід відзначити:


- декілька сот математичних функцій (елементарних, статистичних, спеціальних та ін.);
- можливість розв'язувати систему нелінійних рівнянь, яка складається з більше 200 рівнянь;
- зручні засоби вводу та форматування тексту;
- покращений вивід таблиць;
- різні функції контролю типу даних;
- функції стиску та розширення областей, закритих для модифікації;
- містить оперативний центр ресурсів Resource Centre, який містить самоучитель, електронні книги, систему швидких “шпаргалок” та ін.;
- можливість використання функцій інших систем (Excel, Axum, Matlab та ін.) і фактична інтеграція з ними;
- можливість підготовки складних документів різними користувачами, які працюють в різних місцях, за допомогою операції Collaboratory;
- моделювання роботи складних систем, побудованих із різних блоків, за допомогою системи MathConnex;
- можливість обміну документами через Internet;
- можливість задання розмірних величин в системі СИ;
- підтримка Open GL та Active X;
- можливість зображувати на одному 3D-графіку поверхонь і фігур різного типу;
- можливість повертати 3D-графіки в просторі за допомогою миші;
- анімація 3D-графіків при нажаті клавіші Shift;
- використання покращеної функціональної окраски поверхонь;
- можливість зображувати на одному 3D-графіку різних поверхонь, які перетинаються в просторі;
- громадна кількість засобів форматування графіків;
- можливість запису файла в форматі HTML;
- можливість запису файла в форматах різних версій;
- та ін.

## 1.2. Робота з двовимірною графікою

$$x := -5, -4.9.. 5$$

$$f(x) := \sin(x)$$

$$g(x) := \cos(x)$$



Потрібно звернути увагу, що для того, щоб появився символ “:=” досить нажати клавішу “:”, а символ “..” – клавішу “;”.

Для побудови зображення графіків необхідно вибрати пункт меню “Insert\Graph\X-Y Plot”. Знизу графіка, як бачимо, задається змінна  $x$ , а з лівого краю задаються функції  $f(x)$ ,  $g(x)$ . Самі функції можна безпосередньо вводити зліва. Для того, щоб додати ще одну функцію на графік, потрібно в кінці останньої введеної функції поставити кому.

Для створення анімації використовують пункт меню “View\Animate”.


Для збільшення деякої ділянки графіка необхідно натиснути праву клавішу миші і вибрати пункт “Zoom”.

Для визначати координати точок графіка необхідно натиснути праву клавішу миші і вибрати пункт “Trace”.

Досить широкі можливості має Mathcad при форматуванні графіка. Для цього потрібно натиснути праву клавішу миші і вибрати пункт “Format” або досить два рази натиснути лівою клавішою миші по графіку.


Також в системі Mathcad є можливість побудови графічних зображень в полярній системі координат. Для цього необхідно вибрати пункт меню “Insert\Graph\Polar Plot”.

$$\phi := 0,0.1..2 \cdot \pi$$



Як бачимо вверху в якості параметра використовувалась змінна  $\phi$ . Можна в якості параметра використовувати і функцію, як це показано внизу.

$$\phi := 0,0.1..2 \cdot \pi$$



### 1.3. Робота з тривимірною графікою


В системі Mathcad існують широкі можливості по роботі з тривимірною графікою. Принаймі шість видів поверхонь можна створити і відформатувати у потрібний користувачу вигляд. Це Surface Plot, Data Points, Bar Plot, Contour Plot, Vector Field Plot та Patch Plot. На малюнку показано приклад побудови найбільш поширеного типу поверхні – Surface Plot.

$$x := 0..20$$

$$y := 0..20$$

$$f(x,y) := \sin(x^2 + y^2)$$

$$M_{x,y} := f\left(\frac{x-10}{5}, \frac{y-10}{5}\right)$$



Вікно для форматування тексту вибирається так само, як і в попередньому випадку. Воно складається з восьми закладок:

*Advanced* – установка розширених опцій (перспектива, світлові ефекти, якість друку і т.д.);

*Axis* – установка параметрів координатних осей (тип, товщина і колір ліній осей, число відміток, їх нумерація, масштаб і т.д.);

*Appearance* – установка вигляду графіка (параметрів закраски, ліній і тип точок, які використовуються при побудові поверхні);

*Back Planes* – установка вигляду заднього плану;

*General* – установка загальних характеристик зображення;

*Lighting* – задання умов освітлення та вибір схеми освітлення;

*Special* – задання спеціальних параметрів (контурних ліній, стовпців, інтерполяцій по кольору та ін.);

*Title* – задання титульних надписей та їх параметри.


В системі Mathcad передбачено і параметричне задання графіків. Для прикладу наведено параметричне задання спіралі. Як видно із малюнка, даний графік показано як точковий (Data Points).

$$i := 0..99$$

$$x_i := \cos\left(i \cdot 6 \cdot \frac{\pi}{100}\right)$$

$$y_i := \sin\left(i \cdot 6 \cdot \frac{\pi}{100}\right)$$

$$z_i := i \cdot 6 \cdot \frac{\pi}{100}$$



## 1.4. Використання функцій `root()` та `polyroots()` для знаходження коренів многочленів

Загальний вигляд даних функцій  
*root(вираз, змінна),*  
*polyroots(вираз, змінна).*

Розглянемо приклад використання функції `root()` для знаходження коренів многочлена  $x^3 - 6x^2 + 21x - 52$ :

```
F(x) = x3 - 6x2 + 21x - 52
x := 0                                {задаємо початкове наближення}
x1 := root(F(x), x)                  {знаходимо перший корінь}
x1 = 4
x := 1 + 1 * i                        {задаємо інше початкове наближення}
x2 := root(F(x) / (x - x1), x)        {знаходимо другий корінь}
x2 = 1 + 3,464 * i
x3 := root(F(x) / ((x - x1) * (x - x2)), x) {знаходимо третій корінь}
x3 = 1 - 3,464 * i
```

Розглянемо приклад використання функції `polyroots()` для знаходження коренів цього ж многочлена:

```
V3 = 1  V2 = -6  V1 = 21  V0 = -52    {заносимо у вектор V коефіцієнти даного многочлена}
polyroots(V) =  $\begin{pmatrix} 1 + 3,464 * i \\ 1 - 3,464 * i \\ 1 \end{pmatrix}$     {знаходимо зразу всі корені}
```

Як бачимо, значно простіше використовувати функцію `polyroots()`, ніж `root()`.



## 1.5. Розв’язання системи лінійних алгебраїчних рівнянь

Розглянемо квадратну систему лінійних алгебраїчних рівнянь

$$\begin{cases} a_{11}x_1 + a_{12}x_2 + \dots + a_{1n}x_n = b_1, \\ a_{21}x_1 + a_{22}x_2 + \dots + a_{2n}x_n = b_2, \\ \dots\dots\dots \\ a_{n1}x_1 + a_{n2}x_2 + \dots + a_{nn}x_n = b_n. \end{cases}$$

Складемо матрицю із коефіцієнтів системи  $A = \begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \dots & \dots & \dots & \dots \\ a_{n1} & a_{n2} & \dots & a_{nn} \end{pmatrix}$ .

Також можемо скласти вектор-стовпець  $B = \begin{pmatrix} b_1 \\ b_2 \\ \dots \\ b_n \end{pmatrix}$ .

Зараз можемо записати у матричному вигляді  $Ax = B$ .

Розв’язок можна знайти за формулою  $x = A^{-1} \cdot B$ .

Наприклад,

$$A := \begin{pmatrix} 5 & 3 & 2 \\ 1 & 7 & 8 \\ 9 & 11 & 12 \end{pmatrix} \qquad B := \begin{pmatrix} 5 \\ 1 \\ 7 \end{pmatrix}$$

$$A^{-1} = \begin{pmatrix} -0.071 & -0.25 & 0.179 \\ 1.071 & 0.75 & -0.679 \\ -0.929 & -0.5 & 0.571 \end{pmatrix}$$

$$A^{-1} \cdot B = \begin{pmatrix} 0.643 \\ 1.357 \\ -1.143 \end{pmatrix}$$

## 1.6. Використання функцій `miner()` та `find()` для розв'язання систем нелінійних рівнянь

Розв'яжемо систему нелінійних рівнянь  $\begin{cases} x^2 + y^2 = 5 \\ x + y = 2 \end{cases}$ , використовуючи

команду `find()`.

$$x := 1 \quad y := 1$$

Given

$$x^2 + y^2 = 5$$

$$x + y = 2$$

$$\text{find}(x, y) = \begin{pmatrix} -0.225 \\ 2.225 \end{pmatrix}$$

Як бачимо в дужках команда `find()` містить початкове наближення. Слід зауважити, що команда `find()` використовує точні методи знаходження розв'язку. Але існують системи нелінійних рівнянь, для яких неможливо знайти розв'язок використовуючи точні методи знаходження. Тому в математичній системі Mathcad існує команда `Minerr()`, яка використовує наближені методи знаходження розв'язку.

За допомогою неї розв'яжемо систему нелінійних рівнянь

$$\begin{cases} (x+1)^2 + (y+1)^2 = 5 \\ x + y = 2 \end{cases}$$

$$x := 1 \quad y := 1$$

Given

$$(x+1)^2 + (y+1)^2 = 5$$

$$x + y = 2$$

$$\text{Minerr}(x, y) = \begin{pmatrix} 0.571 \\ 0.615 \end{pmatrix}$$

Слід зауважити, що дана система не розв'язується за допомогою команди `find()`.

## 1.7. Використання функцій `minimize()` та `maximize()` для знаходження найбільших та найменших значень функції

Загальний вигляд даних функцій

*minimize(функція),*

*maximize(функція).*

Розглянемо приклад використання функції `maximize()` для знаходження для знаходження мінімального значення функції  $f(x, y) := x^2 + y^2 + 1$

$$x := 5 \quad y := 5$$

$$f(x, y) := -x^2 - y^2 + 1$$

GIVEN

$$p := \text{maximize}(f, x, y)$$

$$p = \begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

$$f(p_0, p_1) = 1$$

Розглянемо приклад використання функції `minimize()` для знаходження для знаходження мінімального значення функції  $f(x, y) := (x - 1)^2 + (y + 1)^2 + 3$

$$x := 5 \quad y := 5$$

$$f(x, y) := (x + 1)^2 + (y + 3)^2 + 3$$

GIVEN

$$p := \text{minimize}(f, x, y)$$

$$p = \begin{pmatrix} -1 \\ -3 \end{pmatrix}$$

$$f(p_0, p_1) = 3$$

## 1.8. Використання деяких інших команд в Mathcad'i

Для спрощення виразу використовують команду *simplify*. Наприклад,

$$\frac{x^2 - 1}{x + 1} + \frac{x^2 - 1}{x - 1} \text{ simplify} \rightarrow 2 \cdot x$$

Для розкладання  $\sin(kx)$  та  $\cos(kx)$  просто на вираз, який складається лише з  $\sin(x)$  та  $\cos(x)$  використовують команду *expand*. Наприклад,

$$\sin(2x) \text{ expand}, x \rightarrow 2 \cdot \sin(x) \cdot \cos(x) \quad |$$

Перетворення Лапласа здійснюється за допомогою команди *laplace*. Наприклад,

$$\cos(x)^2 \text{ laplace}, x \rightarrow \frac{1}{2} \cdot \frac{s}{(s^2 + 4)} + \frac{1}{(2 \cdot s)}$$

Обернене перетворення Лапласа здійснюється за допомогою команди *invlaplace*. Наприклад,

$$\frac{x}{x + s} \text{ invlaplace}, x \rightarrow \Delta(t) - s \cdot \exp(-s \cdot t)$$

Для розкладання в ряд Тейлора використовується команда *series*. Далі за нею вказується змінна по якій будемо розкладати та степiнь до якої будемо розкладати. Наприклад,

$$e^x \text{ series}, x, 5 \rightarrow 1 + x + \frac{1}{2} \cdot x^2 + \frac{1}{6} \cdot x^3 + \frac{1}{24} \cdot x^4$$

Перетворення Фур'є здійснюється за допомогою команди *fourier*. Наприклад,

$$x^2 \text{ fourier}, x \rightarrow -2 \cdot \pi \cdot \Delta(2, \omega)$$

Обернене перетворення Фур'є здійснюється за допомогою команди *invfourier*. Наприклад,

$$\frac{x + 2}{x + 3} \text{ invfourier}, x \rightarrow \Delta(t) - \frac{1}{2} \cdot \text{li} \cdot \exp(-3 \cdot \text{li} \cdot t) \cdot \Phi(t) + \frac{1}{2} \cdot \text{li} \cdot \exp(-3 \cdot \text{li} \cdot t) \cdot \Phi(-t)$$

## 1.9. Використання команди if()

Загальний вигляд команди

$$\text{if}(\langle\text{умова}\rangle, \langle\text{команда1}\rangle, \langle\text{команда2}\rangle)$$

$\langle\text{команда1}\rangle$  виконується якщо  $\langle\text{умова}\rangle$  вірна, інакше – виконується  $\langle\text{команда2}\rangle$ .

Розглянемо на прикладі деяких задач, як саме працює дана команда.

**Задача 1.** Знайти найбільше з двох чисел.

$$\text{MAX2}(a, b) := \text{if}(a > b, a, b)$$
$$\text{MAX2}(5, 7) = 7$$


**Задача 2.** Знайти найбільше з трьох чисел.

$$\text{MAX3}(a, b, c) := \text{if}[(a > b) \cdot (a > c), a, \text{if}(b > c, b, c)]$$
$$\text{MAX3}(12, 19, 17) = 19$$

*Коментарі.* Як видно з прикладу, дозволяється використовувати вкладення команди if() в саму себе.

**Задача 3.**


Побудуємо деяку функцію  $y=f(x)$

$$x := -2, -1.9.. 2 \quad f(x) := x^2 - 1$$


а) Переробити дану функцію таким чином, щоб на проміжку, де вона менше нуля, вона була рівна 0.

Для цього введемо функцію  $g(x)$  таким чином


$$g(x) := \text{if}(f(x) > 0, f(x), 0)$$



в) Зробити малюнок таким чином, щоб при  $x < 1$ , було намальовано функцію  $y=f(x)$ , а при  $x \geq 1$  – функцію  $y=-f(x)$ .

Для цього введемо функцію  $h(x)$  таким чином


$$h(x) := \text{if}(x \geq 1, f(x), -f(x))$$



б) Зробити малюнок таким чином, щоб проміжку  $x \in [-1, 1]$ , було намальовано функцію  $y=f(x)$ , а інакше – функцію  $y=-f(x)$ .

Для цього введемо функцію  $k(x)$  таким чином


$$k(x) := \text{if}[(x > -1) \cdot (x < 1), f(x), -f(x)]$$



д) Зробити малюнок таким чином, щоб проміжку  $x \notin [-1, 1]$ , було намальовано функцію  $y=f(x)$ , а інакше – функцію  $y=-f(x)$ .

Для цього введемо функцію  $l(x)$  таким чином


$$l(x) := \text{if}[(x < -1) + (x > 1), f(x), -f(x)]$$



е) Зробити малюнок таким чином, щоб проміжку при  $x < -1$ , було намальовано функцію  $y=f(x)$ , а при  $x > 1$  – функцію  $y=-f(x)$ , а інакше пряму рівну 0.

Для цього введемо функцію  $m(x)$  таким чином

$$m(x) := \text{if}(x < -1, f(x), \text{if}(x > 1, -f(x), 0))$$



## 2. Структурне програмування в математичній системі Mathcad

### 2.1. Використання команди if-otherwise

**Задача 1.** Знайти найбільше з двох чисел.

$$\text{MAX2}(a,b) := \begin{cases} a & \text{if } a > b \\ b & \text{otherwise} \end{cases}$$

$$\text{MAX2}(7,9) = 9$$

*Коментарі.* Вертикальна лінія будується за допомогою команди Add Line. Команда otherwise показує, що повинно виконуватись або яке значення повинно присвоюватись, якщо умова біля команди if не виконується.

**Задача 2.** Знайти найбільше з трьох чисел.

$$\text{MAX3}(a,b,c) := \begin{cases} a & \text{if } (a > b) \wedge (a > c) \\ & \text{otherwise} \\ & \begin{cases} b & \text{if } b > c \\ c & \text{otherwise} \end{cases} \end{cases}$$

$$\text{MAX3}(17,12,19) = 19$$


**Задача 3.** Побудувати функцію  $k(x)$ , яка задається наступним чином

$$k(x) = \begin{cases} \frac{1}{|x|}, & x \in (-\infty, -2) \cup (2, +\infty) \\ -\frac{3}{2} \cdot x - \frac{5}{2}, & x \in (-2, -1) \\ x, & x \in (-1, 1) \\ -2, & \text{інакше} \end{cases}$$

$$k(x) := \begin{cases} \frac{1}{|x|} & \text{if } (x > 2) \vee (x < -2) \\ x & \text{if } -1 < x < 1 \\ -\frac{3}{2} \cdot x - \frac{5}{2} & \text{if } (x < -1) \wedge (x > -2) \\ -2 & \text{otherwise} \end{cases}$$

$$x := -6, -5.9 .. 6$$





**Задача 4.** Створити одиничну матрицю розмірності  $n \times n$ .

$$I(n) := \begin{cases} \text{for } i \in 0..n-1 \\ \quad \text{for } j \in 0..n-1 \\ \quad \quad \begin{cases} m_{i,j} \leftarrow 1 & \text{if } i = j \\ m_{i,j} \leftarrow 0 & \text{otherwise} \end{cases} \end{cases} \\ m$$

$$I(3) = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

**Задача 5.** Створити матрицю, яка буде складатись з вкладених в неї матриць, які складаються з одиничок.

$$A(n) := \begin{cases} B_0 \leftarrow \text{identity}(1) \\ \text{for } k \in 1..n \\ \quad \begin{cases} B_k \leftarrow \text{augment}(B_{k-1}, B_{k-1}) & \text{if } \text{mod}(k, 2) = 1 \\ B_k \leftarrow \text{stack}(B_{k-1}, B_{k-1}) & \text{otherwise} \end{cases} \end{cases} \\ B$$

$$A(5) = \begin{pmatrix} \{1,1\} \\ \{1,2\} \\ \{2,2\} \\ \{2,4\} \\ \{4,4\} \\ \{4,8\} \end{pmatrix} \quad \begin{aligned} A(5)_0 &= ( 1 ) \\ A(5)_1 &= ( 1 \ 1 ) \\ A(5)_2 &= \begin{pmatrix} 1 & 1 \\ 1 & 1 \end{pmatrix} \\ A(5)_3 &= \begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 \end{pmatrix} \end{aligned}$$

*Коментарі.* Команда `augment(A, B, C, ...)` поверне матрицю, яка буде складатись з матриць A, B, C, ... складених зліва направо. Команда `stack(A, B, C, ...)` поверне матрицю, яка буде складатись з матриць A, B, C, ... складених зверху вниз.

**Задача 6.** Побудувати вектор, який буде складатися з цифр від 1 до n, які діляться на a або b.

$$S(n, a, b) := \begin{array}{l} | j \leftarrow 0 \\ | \text{for } k \in 1..n \\ | \quad \text{if } (\text{mod}(k, a) = 0) \vee (\text{mod}(k, b) = 0) \\ | \quad \quad | v_j \leftarrow k \\ | \quad \quad | j \leftarrow j + 1 \\ | \quad \quad v \end{array}$$

$$S(10, 2, 3)^T = ( 2 \ 3 \ 4 \ 6 \ 8 \ 9 \ 10 )$$

**Задача 7.** Написати функцію, яка б зразу повертала мінімальний та максимальний елементи вказаного вектору-масиву.

$$\text{minmax}(v, n) := \begin{array}{l} | \text{min} \leftarrow v_0 \\ | \text{max} \leftarrow v_0 \\ | \text{for } i \in 0..n-1 \\ | \quad | \text{min} \leftarrow v_i \text{ if } \text{min} > v_i \\ | \quad | \text{max} \leftarrow v_i \text{ if } \text{max} < v_i \\ | \quad \begin{pmatrix} \text{min} \\ \text{max} \end{pmatrix} \end{array}$$

$$v := \begin{pmatrix} 11 \\ 15 \\ 8 \\ 17 \\ 9 \end{pmatrix} \quad \text{minmax}(v, 5) = \begin{pmatrix} 8 \\ 17 \end{pmatrix}$$

**Задача 8.** Написати програму, яка б шукала індекси елемента заданої матриці, який точно рівний вказаному.

```

FindIndex(M, key) :=
  r ← rows(M) - 1
  c ← cols(M) - 1
  i ← 0
  j ← 0
  while (j ≤ c) · (Mi,j ≠ key)
    i ← i + 1 if i < r
    otherwise
      i ← 1
      j ← j + 1
  ( i )
  ( j ) if j ≤ c
  0 otherwise

```

$i := 0..99 \quad j := 0..99 \quad N_{i,j} := \text{rnd}(100) \quad \text{max}(N) = 99.99$

$\text{FindIndex}(N, \text{max}(N)) = \begin{pmatrix} 54 \\ 22 \end{pmatrix}$

*Коментарі.* Команда  $\text{rnd}(N)$  повертає дійсне число з діапазону від 1 до  $N$ . Команда  $\text{max}(A)$  повертає найбільший елемент матриці  $A$ .

## 2.2. Використання команди for

**Задача 1.** Знайти суму чисел від 1 до n.

$$\text{sum}(n) := \left| \begin{array}{l} s \leftarrow 0 \\ \text{for } i \in 1..n \\ \quad s \leftarrow s + i \end{array} \right.$$

$$\text{sum}(5) = 15$$

**Задача 2.** Знайти n!.

$$F(n) := \left| \begin{array}{l} f \leftarrow 1 \\ \text{for } i \in 1..n \\ \quad f \leftarrow f \cdot i \\ f \end{array} \right.$$

$$F(6) = 720$$

**Задача 3.** Знайти n-те по порядку число Фібоначчі .

$$G(n) := \left| \begin{array}{l} g \leftarrow 1 \text{ if } n \leq 1 \\ \text{otherwise} \\ \quad \left| \begin{array}{l} h \leftarrow 1 \\ g \leftarrow 1 \\ \text{for } k \in 2..n \\ \quad \left| \begin{array}{l} \text{tmp} \leftarrow h + g \\ h \leftarrow g \\ g \leftarrow \text{tmp} \end{array} \right. \end{array} \right. \\ g \end{array} \right.$$

$$G(7) = 21$$

**Задача 4.** Записати матрицю  $M$  в один довгий вектор стовпець, починаючи з останнього стовпця.

$M := \text{identity}(3)$

$$\text{flatten}(M) := \left| \begin{array}{l} v \leftarrow M \langle \text{cols}(M)-1 \rangle \\ \text{for } j \in \text{cols}(M) - 2..0 \quad \text{if } \text{cols}(M) > 1 \\ \quad v \leftarrow \text{stack}(v, M \langle j \rangle) \\ v \end{array} \right.$$

$\text{flatten}(M)^T = ( 0 \ 0 \ 1 \ 0 \ 1 \ 0 \ 1 \ 0 \ 0 )$

*Коментарі.* Команда  $\text{identity}(n)$  – повертає одиничну матрицю розмірності  $n$ .

**Задача 5.** Знайти добуток елементів вектора  $v$ .

$$v := \pi \ e \ \frac{1 + \sqrt{5}}{2}^T$$

$$\text{prod}(v) := \left| \begin{array}{l} p \leftarrow 1 \\ \text{for } i \in v \\ \quad p \leftarrow p \cdot i \\ p \end{array} \right.$$

$\text{prod}(v) = 13.818$

**Задача 6.** Створити вектор, який би містив числа з вказаного діапазону з певним кроком.

$$\text{vec}(\text{low}, \text{upp}, \text{incr}) := \left| \begin{array}{l} \text{ind} \leftarrow 0 \\ \text{for } i \in \text{low}, \text{low} + \text{incr}.. \text{upp} \\ \quad \left| \begin{array}{l} v_{\text{ind}} \leftarrow i \\ \text{ind} \leftarrow \text{ind} + 1 \end{array} \right. \\ v \end{array} \right.$$

$\text{vec}(1, 2, 0.2)^T = ( 1 \ 1.2 \ 1.4 \ 1.6 \ 1.8 \ 2 )$

*Коментарі.* Змінні low, upr – показують нижню і верхню межі діапазону, а змінна incr показує крок.

**Задача 7.** Задана матриця  $A[1..M, 1..N]$ . Вона заповнена 0 і 1. Знайти в ній площу найбільшого квадрата, який складається з 1.

$$A := \begin{pmatrix} 1 & 0 & 1 & 1 & 0 \\ 1 & 1 & 1 & 1 & 1 \\ 1 & 0 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 0 & 1 \end{pmatrix}$$

$$\text{sqr1}(A, M, N) := \begin{array}{l} \text{for } i \in 0.. M - 1 \\ \quad B_{i,0} \leftarrow A_{i,0} \\ \text{for } j \in 0.. N - 1 \\ \quad B_{0,j} \leftarrow A_{0,j} \\ \text{for } i \in 1.. M - 1 \\ \quad \text{for } j \in 1.. N - 1 \\ \quad \quad \left| \begin{array}{l} B_{i,j} \leftarrow 0 \text{ if } A_{i,j} = 0 \\ B_{i,j} \leftarrow \min(B_{i-1,j}, B_{i,j-1}, B_{i-1,j-1}) + 1 \text{ otherwise} \end{array} \right. \\ \quad \max(B)^2 \end{array}$$

$$\text{sqr1}(A, 5, 5) = 9$$

## 2.2. Використання команди while

**Задача 1.** Написати функцію, яка буде знаходити корінь квадратний із заданого числа із вказаною точністю.

a)

$$\text{sqrt}(a, \varepsilon) := \left| \begin{array}{l} \text{estsqrt} \leftarrow 1 \\ \text{while } \left| \text{estsqrt}^2 - a \right| \geq \varepsilon \\ \quad \text{estsqrt} \leftarrow \frac{1}{2} \cdot \left( \text{estsqrt} + \frac{a}{\text{estsqrt}} \right) \end{array} \right.$$

$$\text{sqrt}(37, .00001) = 6.082762537585$$

б)

$$\text{sroot}(a, \varepsilon) := \left| \begin{array}{l} \text{estsqrt} \leftarrow 1 \\ \text{while } 1 \\ \quad \left| \begin{array}{l} \text{estsqrt} \leftarrow \frac{1}{2} \cdot \left( \text{estsqrt} + \frac{a}{\text{estsqrt}} \right) \\ \text{break if } \left| \text{estsqrt}^2 - a \right| < \varepsilon \end{array} \right. \\ \text{estsqrt} \end{array} \right.$$

$$\text{sroot}(37, .00001) = 6.082762537585$$

*Коментарі.* Команда break перериває роботу циклу, як тільки виконується умова, яка стоїть біля неї.

**Задача 2.** Знайти перший по порядку елемент вектора  $v$ , який по абсолютній величині більший вказаного.

$$j := 0..2500 \quad v_j := \sin(j)$$

$$t(v, \text{thres}) := \left| \begin{array}{l} j \leftarrow 0 \\ \text{while } |v_j| \leq \text{thres} \\ \quad j \leftarrow j + 1 \\ \quad \left( \begin{array}{c} j \\ v_j \end{array} \right) \end{array} \right.$$

$$t(v, 0.999990) = \begin{pmatrix} 11 \\ -0.99999021 \end{pmatrix}$$

$$t(v, 0.999991) = \begin{pmatrix} 2474 \\ -0.99999112 \end{pmatrix}$$

**Задача 3.** Знайти перший по порядку елемент вектора  $v$ , який рівний вказаному.

$$j := 0..500 \quad w_j := \text{ceil}(1000 \cdot \cos(j))$$

$$\text{search}(v, x) := \begin{array}{l} j \leftarrow 0 \\ \text{while } v_j \neq x \\ \quad j \leftarrow j + 1 \\ \left( \begin{array}{c} j \\ v_j \end{array} \right) \end{array}$$

$$\text{search}(w, 400) = \begin{pmatrix} 397 \\ 400 \end{pmatrix} \quad \text{search}(w, 700) = \begin{pmatrix} 150 \\ 700 \end{pmatrix}$$

**Задача 4.** Знайти перший елемент матриці, модуль якого відрізняється від 0 на не більше, ніж на вказану величину.

$$i := 0..15 \quad j := 0..15 \quad m_{i,j} := \sin(i) + \cos(j)$$

	0	1	2	3	4	5
0	1	0.54	-0.416	-0.99	-0.654	0.284
1	1.841	1.382	0.425	-0.149	0.188	1.125
2	1.909	1.45	0.493	-0.081	0.256	1.193
3	1.141	0.681	-0.275	-0.849	-0.513	0.425
4	0.243	-0.217	-1.173	-1.747	-1.41	-0.473
5	0.041	-0.419	-1.375	-1.949	-1.613	-0.675
6	0.721	0.261	-0.696	-1.269	-0.933	$4.247 \cdot 10^{-3}$
7	1.657	1.197	0.241	-0.333	$3.343 \cdot 10^{-3}$	0.941
8	1.989	1.53	0.573	$-6.342 \cdot 10^{-4}$	0.336	1.273
9	1.412	0.952	$-4.028 \cdot 10^{-3}$	-0.578	-0.242	0.696
10	0.456	$-3.719 \cdot 10^{-3}$	-0.96	-1.534	-1.198	-0.26
11	$9.793 \cdot 10^{-6}$	-0.46	-1.416	-1.99	-1.654	-0.716
12	0.463	$3.729 \cdot 10^{-3}$	-0.953	-1.527	-1.19	-0.253
13	1.42	0.96	$4.02 \cdot 10^{-3}$	-0.57	-0.233	0.704
14	1.991	1.531	0.574	$6.149 \cdot 10^{-4}$	0.337	1.274
15	1.65	1.191	0.234	-0.34	$-3.356 \cdot 10^{-3}$	0.934



$$z(m, \delta) := \left| \begin{array}{l} k \leftarrow 0 \\ \text{while } k < \text{rows}(m) \\ \quad \left| \begin{array}{l} i \leftarrow k \\ j \leftarrow 0 \\ \text{while } (i \geq 0) \cdot (|m_{i,j}| \geq \delta) \\ \quad \left| \begin{array}{l} i \leftarrow i - 1 \\ j \leftarrow j + 1 \end{array} \right. \\ k \leftarrow k + 1 \text{ if } i < 0 \\ \text{break if } i \geq 0 \end{array} \right. \\ \left( \begin{array}{c} i \\ j \\ m_{i,j} \end{array} \right) \end{array} \right.$$

$$z(m, 0.0001) = \left( \begin{array}{c} 11 \\ 0 \\ 9.793 \times 10^{-6} \end{array} \right)$$

*Коментарі.* Команда `rows(m)` повертає кількість рядків матриці `m`. Зверніть знову увагу, що нумерація в масивах починається з нульового індексу.

**Задача 5.** Знайти  $n!$ .

a)

$$F(n) := \left| \begin{array}{l} f \leftarrow 1 \\ \text{while } n \leftarrow n - 1 \\ \quad f \leftarrow f \cdot (n + 1) \\ f \end{array} \right.$$

$$F(6) = 720$$

6)

```
F(n) := | f ← n
        | while 1
        |   | f ← f · (n - 1)
        |   | n ← n - 1
        |   | break if n = 1
        | f
```

$F(6) = 720$

### 3. ПЕРЕЛІК ЛАБОРАТОРНИХ РОБІТ

#### 3.1. Лабораторна робота №1

##### *Прості операції та двовимірна графіка*

1) Обчислити:

а)  $\int_0^{10} \sqrt{x^2 - 5x + 1} dx$ ;

б)  $\sum_{i=1}^{15} i^3$ ;

в)  $\lim_{x \rightarrow 0} \frac{\sin x}{x}$ ;

г)  $\prod_{i=1}^{10} i$ ;

д) Знайти обернену матрицю до деякої матриці  $M$  розмірністю  $3 \times 3$ .

е) Розв'язати систему лінійних однорідних рівнянь, записаних у вигляді  $Ax=B$ . Масиви  $A$  і  $B$  зчитуються із файлу.

2) В декартовій системі координат побудувати графіки функцій:

$$f_1(x) = \sin(x); \quad f_2(x) = \cos(x); \quad f_3(x) = \frac{\operatorname{tg}(x)}{x}.$$

(Insert\Graph\X-Y Plot)

Зробити наступні налаштування:

- Значення  $x$  змінювалось від  $-5$  до  $5$  з кроком  $0.1$ ;
- Вісь ординат задавалась проміжком  $[-2, 2]$ ;
- Задати осі у вигляді хреста (Format\X-Y Axes\Crosed);
- Задати однаковий масштаб по осям (Equal Scales);
- Задати першу функцію прямокутниками, другу – точками, третю – суцільною лінією. Всі різного кольору. (Format\Traces).
- Задати титульний надпис "Графіки функцій" і зробити надписи по осям координат "Вісь X" і "Вісь Y";
- Показати, яка функція яким типом лінії задана;
- Нанести зелену сітку;
- Подрібнити зелену сітку;
- Вивести значення функцій  $f_1(x)$ ,  $f_2(x)$ ,  $f_3(x)$  на екран.

Уміти:

- Проглядати деякі ділянки графіка (Zoom);
- Уміти визначати координати точок графіка функції (Trace);
- Робити анімацію (View\Animate);
- Вставляти даний графік у документ текстового редактора Word.

3) Побудувати графіки функції у полярній системі координат:

$$f_1(W) = \sin(2W), \quad f_2(W) = \sin(3W), \quad f_3(W) = \cos(3W), \quad f_4(W) = \cos(4W).$$

$$(W := 0, 0.1\pi \dots 2\pi, \quad R_{\min} := 0, \quad R_{\max} := 1)$$

Зробити наступні налаштування:

- задати вісь у вигляді кола;
- задати титульну надпись "Полярна система координат";
- задати пунктирною лінією другий графік;

- показати, які значення приймає кожна функція окремо;
- показати, яка функція яким типом лінії задана.

Замітка: Всі три попередні завдання зробити у різних файлах.  
В новому файлі створити гіперпосилання на кожне завдання.  
(Insert\Hyperlink)

### 3.2. Лабораторна робота №2

*Використання 3-х мірної графіки в математичній системі MathCAD*

1) Побудувати графік функції  $f(x,y)=\cos(x^2+y^2)$ .  
(Surface Plot)

Уміти робити наступні налаштування:

- задати  $x \in [-20, 20]$ ,  $y \in [-20, 20]$ ;
- задати назву графіка;
- задати однаковий масштаб по осям;
- задати осі різними кольорами;
- встановити, щоб не показувало номери поділок осей  $x$  і  $y$ .
- встановити, щоб не показувало невидимі частини даної фігури;
- встановити паралелепіпед, який буде обмежувати нашу фігуру;
- за допомогою миші перевернути його вниз головою;
- заставити поверхню рухатись;
- нанести на нашу фігуру деяку текстуру;
- задати перспективу;
- задати туман;
- встановити площину XY з розбиттям, закрашену в жовтий колір;
- встановити інші площини;
- вставити даний графік у документ текстового редактора Word.

2) Побудувати точковий графік функції, яка задана параметрично:  
 $N=100 \quad i=0..N-1$

$$x_i = \cos\left(\frac{i}{N} \cdot 6\pi\right) \quad y_i = \sin\left(\frac{i}{N} \cdot 6\pi\right) \quad z_i = \frac{i}{N} \cdot 6\pi$$

Встановити налаштування аналогічні до першого завдання.

3) Побудувати інші 3D графіки для функції  $f(x,y)=\sin(x^2+y^2)$ .  
Оформити дану роботу аналогічно лабораторній роботі №1.

### 3.3. Лабораторна робота №3

*Розв'язання задач в системі MathCAD*

- 1) Дано координати двох клітин на шахматній досці  $(x1,y1)$  та  $(x2,y2)$ .
  - а) перевірити чи будуть вони одного кольору;
  - б) якщо на них розмістити ферзі, то чи будуть бити вони один одного;
  - в) аналогічно для коней.
- 2) Дано більярдний стіл розмірністю  $(M \times N)$  ( $M, N$  – натуральні числа). З нижньої лівої лузи вилітає кулька під кутом  $45^\circ$ . Через скільки ударів вона попаде знову в лузу і в яку саме.
- 3) Задана матриця  $A[1..M, 1..N]$ . Вона заповнена 0 і 1. Знайти в ній площу найбільшого квадрата, який складається з 1.
- 4) КІЛЬКІСТЬ РОЗБИТТІВ. Ціле додатне число можна зобразити сумою його частин, що називається розбиттям. Наприклад, число 4 можна зобразити так: 4; 3+1; 2+2; 2+1+1; 1+1+1+1.  $P(n)$  – кількість розбиттів,  $P(4)=5$ . Для заданого  $n$  визначити значення функції  $P(n)$ .
- 5) Кожну задачу розмістити в різних файлах. Виклик кожного файлу зробити у вигляді гіперпосилань.

### 3.4. Лабораторна робота №4

*Використання математичної системи MathCAD4 в статистиці, розв'язання систем нелінійних рівнянь, оптимізації*

- 1) Побудувати лінію регресії  $y = ax + b$  для вибору  $x$  і  $y$ , які зчитуються з файлу. Знайти при цьому середні значення  $\bar{x}$  і  $\bar{y}$ , середньоквадратичні відхилення  $\delta x$  і  $\delta y$ , коефіцієнт кореляції і коваріації. Вивести таблиці, які необхідні, коли обчислення роблять вручну. Побудувати гістограму для цих наборів.
- 2) Навчитись користуватись функціями root і polyroots. Загальний вигляд root(вираз, змінна), polyroots(вираз, змінна).
- 4) Розв'язати с-му нелінійних рівнянь, використовуючи функції FIND, MINERR.
- 4) Навчитись користуватись функціями MINIMIZE, MAXIMIZE.

### 3.5. Лабораторна робота № 5

*Використання системи MathCad в різних розділах математики*

1. Загальна частина.
  - а) Спростити вираз:  
$$-\frac{5}{x} + \frac{5}{(x-1)} - \frac{5}{(x-1)^2} + \frac{6}{(x-1)^3} - \frac{4}{(x-1)^4}$$
 (використовуючи simplify).
  - б) Розкласти в ряд Тейлора функцію  $f(x)=\sin x$  (викор. series)
  - в) здійснити перетворення Лапласа функції  $f(x)=(\sin x)^2$  (викор. laplace)

г) знайти комплексні корені р-ня  $x^3 - 6x^2 + 21x - 52 = 0$

д) здійснити розклад  $\sin$  кратних кутів( $\sin(5x)\dots$ )

(викор. expand)

е) розв'язати систему диф. р-нь

$$f'_{x_0}(x_0, x_1) = -x_0 - x_1 - (x_0^2 + x_1^2) \cdot x_0$$

$$f'_{x_1}(x_0, x_1) = -x_1 + x_0 - (x_0^2 + x_1^2) \cdot x_1$$

(викор. rkfixed)

Побудувати графік фазових коливань.

є) здійснити лін. (lspline,interp)і сплайнову інтерполяцію для набору

X	1	2	3	4	4.5	5
Y	1	3	4	2.5	2	5

Побудувати графіки.

## 2. Додаткова частина.

А) дана матриця дійсних чисел  $A(m \times n)$ . Робот знаходиться в клітині (1,1). Він може рухатись або вниз, або вправо. Яку максимальну суму він може назбирати, попавши в клітину (m,n). Вказати також і шлях.

Б) Дано  $N^2$  натуральних чисел  $1, 2, 3, \dots, N^2$ . Розбити їх на  $N$  груп таким чином, щоб

- в кожній групі було по  $N$  чисел;
- сума чисел у кожній групі була однаковою.

В) Заданий квадратний аркуш паперу. Розграфлений  $N \times N$  квадратиків, де  $N=2^m$  ( $m$  – вхідний параметр). Квадратики пронумеровані від 1 до  $N^2$  вздовж рядків на I, II, III, IV. Четвертинки накладають: II на I, III на II, IV на III. Одержану стопку знову розрізають, ... Процедуру розрізування і накладання повторюють до  $m$  разів. На кінцевому етапі одержують стопку (послід-ть)  $1 \times N^2$  квадратиків (рахують знизу). Скласти програму, яка визначить в кінцевій стопці номер, в початковій нумерації,  $K$ -го порядку квадрата.

## Рекомендована література

1. Дьяконов В.П., Абраменкова И.В. Mathcad 8 PRO в математике, физике и Internet. – М.: Нолидж. – 1999. – 512 с.
2. Дьяконов В.П., Абраменкова И.В. Mathcad 7 в математике, физике и Internet. – М.: Нолидж. – 1998. – 352 с.
3. Дьяконов В.П. Справочник по Mathcad Plus 7.0 PRO. – М.: СК-ПРЕСС. – 1998. – 352 с.
4. Дьяконов В.П. Справочник по системе символьной математики Derive. – М.: СК-ПРЕСС. – 1998. – 256 с.
5. Дьяконов В.П. Системы символьной математики Mathematica 2 и Mathematica 3. – М.: СК-ПРЕСС. – 1998. – 320 с.
6. Дьяконов В.П. Математическая система Maple V R3/R4/R5. – М.: Солон. – 1998. – 398 с.
7. Дьяконов В.П. Справочник по применению системы PC MatLAB. – М.: Наука. Физматлит. – 1993. – 112 с.
8. Дьяконов В.П. Mathcad прорывается в Windows // Монитор-Аспект, 1993. – №2. – с. 42.
9. Очков В.Ф. Mathcad 7 PRO для студентов и инженеров. – М.: Компьютер Пресс. – 1998. – 384 с.
10. Mathcad 6.0 PLUS // Перевод с англ. руководства пользователя. – М.: Филинь. – 1996. – 712 с.

## ЗМІСТ

Вступ	1
<b>1. Загальні можливості математичної системи Mathcad</b>	<b>2</b>
1.1. Загальні відомості	2
1.2. Робота з двовимірною графікою	4
1.3. Робота з тривимірною графікою	6
1.4. Використання функцій root() та polyroots() для знаходження коренів многочленів	8
1.5. Розв'язання системи лінійних алгебраїчних рівнянь	9
1.6. Використання функцій miner() та find() для розв'язання систем нелінійних рівнянь	10
1.7. Використання функцій minimize() та maximize() для знаходження найбільших та найменших значень функції	11
1.8. Використання деяких інших команд в Mathcad'і	12
1.9. Використання команди if()	13
<b>2. Структурне програмування в математичній системі Mathcad</b>	<b>16</b>
2.1. Використання команди if-otherwise	16
2.2. Використання команди for	20
2.3. Використання команди while	23
<b>3. Перелік лабораторних робіт</b>	<b>27</b>
3.1. Лабораторна робота №1	27
3.2. Лабораторна робота №2	28
3.3. Лабораторна робота №3	29
3.4. Лабораторна робота №4	29
3.5. Лабораторна робота №5	29
<b>Рекомендована література</b>	<b>31</b>